
clear all 
******************************************************************************* 
********************************** EXAMPLES *********************************** 
******************************************************************************* 
 
use "C:\Users\Cyrus\Desktop\ADQUANT - 020720\GSS2016.dta" 
 
 
**************************** LOGIT Model COmparison *************************** 
** Coding of Variables *** 
recode attend (7 8 =1 "Weekly")(.d=.) (else=0 "Non-Weekly"), gen(wkattnd) 
  
  * Check 
  tab wkattnd 
 
* Political Ideology 
recode polviews (1 2 3=1 "libs.") (4=0 "mods") (5 6 7=2 "cons."), gen (pview)  
   
  * Check 
  tab pview  
   
   
** recode degree for advanced and BA 
recode degree (3=1 "BA")(4=2 "Advance") (0 1 2=0 "No Col.") (.n=.), gen(cdeg) 
 
  * Check 
  tab cdeg 
 
 
******************************************************************************* 
***************************** MODEL/GROUP COMPARISON ************************** 
******************************************************************************* 
eststo clear 
** Full 
eststo: logit wkattnd i.pview i.sex i.cdeg i.race c.age  
 
** Men 
eststo: logit wkattnd i.pview i.cdeg i.race c.age if sex==1 
 
** Women 
eststo: logit wkattnd i.pview i.cdeg i.race c.age if sex==2 
 
** Interaction comparions 
eststo:logit wkattnd i.pview##i.sex i.cdeg i.race c.age  
 
  ** Comparison Table 
  esttab, b(%9.2f) se(%9.2f) aic bic nonumber nodepvars varwidth(15)         /// 
    drop (0.pview 1.sex 0.cdeg 1.race 0.pview#1.sex 1.pview#1.sex            /// 
   2.pview#1.sex 0.pview#2.sex)                                           /// 
 order(1.pview 2.pview 2.sex 1.pview#2.sex 2.pview#2.sex )                /// 
    coeflabel( 1.pview "Liberals" 2.pview "Conservatives" 1.cdeg "Bachelors" /// 
      2.cdeg "Advanced Deg." 2.race "Black" 3.race "Other Race" age "Age"    /// 
   2.sex "Female" _cons "Intercept" 1.pview#2.sex "Libs*Female"           /// 
   2.pview#2.sex "Cons*Female")                                           /// 
    mtitle("Full" "Men" "Women" "Interaction")    
 
 
** Interaction comparions 
logit wkattnd i.pview##i.sex i.cdeg i.race c.age  
 
 ** To test the differences in the probabilities first we post them so we can 
 ** recall for testing 
 margins pview#sex, post   



  
 ** Plotting is a good way to eyeball difference 
 marginsplot,recast(bar) x(pview sex) xlabel(, angle(30)) 
  
   ** Formal statistical tests of difference - Moderates by gender 
   test 0.pview#1.sex=0.pview#2.sex 
  
   ** Formal statistical tests of difference - Liberal by gender 
   test 1.pview#1.sex=1.pview#2.sex 
     
   ** Formal statistical tests of difference - Conservatives by gender 
   test 2.pview#1.sex=2.pview#2.sex    
     
** What else can we test under these conditions? ** 
  
    ** Formal statistical tests of difference - Mod v lib for males nder 
    test 0.pview#1.sex=1.pview#1.sex    
  
** How do we report this? 
  
  eststo clear 
  ** Formal statistical tests of difference - no college by gender 
  eststo: logit wkattnd i.pview##i.sex i.cdeg i.race c.age  
   
    ** Margins for testing 
 margins pview#sex, post 
  
 ** Formal statistical tests of difference - no college by gender 
 test 0.pview#1.sex=0.pview#2.sex 
      estadd scalar chi1  =  r(chi2) 
   estadd scalar df1 =  r(df) 
   estadd scalar p1  =  r(p) 
    
  ** Saving it all 
  eststo 
     
** Table 
esttab, b(%9.2f) se(%9.2f) aic bic nonumber nodepvars varwidth(15)           /// 
    drop (0.pview 1.sex 0.cdeg 1.race )                                      /// 
 order(1.pview 2.pview 2.sex 0.pview#1.sex 0.pview#2.sex  1.pview#1.sex   /// 
   1.pview#2.sex 2.pview#1.sex 2.pview#2.sex)                             /// 
    coeflabel( 1.pview "Liberals" 2.pview "Conservatives" 1.cdeg "Bachelors" /// 
      2.cdeg "Advanced Deg." 2.race "Black" 3.race "Other Race" age "Age"    /// 
   2.sex "Female" _cons "Intercept" 0.pview#1.sex "Mods*Male"             /// 
   0.pview#2.sex "Mods*Female" 1.pview#1.sex "Libs*Male" 1.pview#2.sex    /// 
   "Libs*Female" 2.pview#1.sex "Cons*Male" 2.pview#2.sex "Cons*Female")   /// 
    mtitle("Full" "Probablities")  scalar(chi1 df1 p1) 
 
  
** Age dif by race  
logit wkattnd i.pview i.sex i.cdeg i.race##c.age##c.age  
 
  ** Margins 
  margins race, at(age=(20(10)90)) 
   
    **Plot it 
    marginsplot, recastci(rarea) legend(row(1))   
   
  ** Margins 
  margins r.race if race!=1, at(age=(20(10)90)) 
   
      **Plot it 
    marginsplot, recastci(rarea) legend(row(1)) yline(0, lw(thick)) 


